State_Variable_OTA_1KHz

```
*   dsauersanjose@aol.com
*   www.idea2ic.com
*   4/15/08
```




```
*          Set  A0  = 1 and scale s to 1KHz
*          Then R2  = Q and  s  = 2*PI*1KHz

.OPTIONS  GMIN=1e-12    METHOD=trap   srcsteps = 1  gminsteps = 1
*======== ====== ====== ====== ====== ====== ====== ====== ======
V_IN      VIN    0      AC     1      DC     0
R0        VIN    SUMN   10k
R1        SUMN   HP     10k
R2        SUMN   BP     100k
R3        SUMN   LP     10k
XOPA1     SUMN   0      HP     OPA
XOTAS1    HP     BP            OTA_S
XOTAS2    BP     LP            OTA_S
.ac       dec    50     10     10k

*==OTAs_Can_Perform_The_Exact_Same_Function====
.control
run
plot  db(bp)  db(hp) db(lp)   title  StateVariable_Q_10
*==Q_Is_Still_Defined_By_FeedBack===========
alter  R2     resistance =   10k
run
plot db(bp)  db(hp) db(lp)   title  StateVariable_Q_1
.endc
```




```
*  http://www.idea2ic.com/PlayWithJavascript/R_C_Freq.html
```



```
.SUBCKT OPA   INP     INN     OUT
EGN1          O1      0       INP     INN     -1
EGN2          OUT     0       IP      0       -1000000
R1            O1      IP      10k
C1            OUT     IP      10p
.ends

.SUBCKT OTA_S IN      OUT
QN1           VC1     IN      VE1     NPNP
QN2           VC2     0       VE1     NPNP
QP1           VC1     VC1     VCC     PNPP
QP2           VC2     VC1     VCC     PNPP
IB            VE1     0               5.2u
VCC           VCC     0               DC      2
EGN2          OUT     0       VC2     0       +1
C1            VC2     0               .01592u
.ends

.model    NPNP   NPN(   BF=2100 VAF=216 )
.model    PNPP   PNP(   BF=2100 VAF=210)


.end

======================END_OF_SPICE=======================
```

A real OTA and capacitor with a voltage buffer can replace the integrator with no effect



Now R1->R3 independently controls Q
The frequency however now is independently control by bias current IB of the OTA.


Graph 121 - ac323: StateVariable_Q_10
dB
db(bp)
db(lp)
db(hp)
20
0
-20
-40
-60
-80
-100
decibel
10
100
10^3
10^4
frequency Hz